

Excel – Formula and Function Tips

By Neale Blackwood

www.a4accounting.com.au

a4@iinet.net.au



Table of Contents

Introduction	3
Quick Tips	4
Plus Sign	4
Numeric Keypad	4
Using the Arrow Keys	4
AutoComplete	4
Last Bracket	4
AutoSum	5
F4 Key	5
Copy Down	6
Calculation Sequence	7
Using Brackets	7
Brackets Colour	7
Formula Colours	8
Large Range Tips	9
Whole Columns	9
Selecting Large Ranges	9
Typing Shortcuts	9
Format As Table	10
Include Inserted Rows	12
Other Sheet Ranges Tip	13
Conclusion	14

Introduction

In my previous webinars, covering lots of other functions, I have included numerous shortcuts and tips to speed up formula entry. This session will bring all of those shortcuts together in one session.

Entering formulas can be a tedious task but there are many shortcuts built into Excel that can make the process a lot easier and quicker.

Most of these involve the keyboard but a few of them also utilise the mouse.

I use range names in most of my spreadsheets however this session will not include any range names. I recommend using range names as they can speed up formula entry even more. I will share other techniques that can help speed up formula entry.

Practice

Some of these techniques can take a little bit of practice to master but they are worth the time as they will ultimately save you so much time in the future. They become second nature quite quickly.

Function Arguments

When discussing functions in Excel I use the term arguments. This is the term used to describe the parts of the function. Some of Excel's functions do not use arguments e.g.

=TODAY()

This doesn't require an argument as it enters today's date in a cell. It updates whenever the file is opened or calculated.

Arguments are separated by commas and are between the brackets of the function.

The correct term for the brackets is actually parentheses, but I use the more common term of brackets, which is easy to write and say.

Some of Excel's functions only have one argument. Other functions have multiple arguments separated by commas.

Some arguments are optional and these are displayed with square brackets around the argument name when you see the arguments listed.

Functions

I won't be going into detail about the functions used in this session. Please refer to the Excel Help system or Google if you need more information on a particular function.

Basics

This session does cover some basic techniques, but in my experience not everyone is aware of the basics. Many people are self-taught and haven't always learned the easiest ways to create formulas.

Techniques

When demonstrating and explaining some techniques I will do them slowly, but I will also show them to you at normal speed, so you can see how fast they can be done.

Quick Tips

Plus Sign

If you have attended any of my previous formula or function webinars then you will have seen me use the plus sign to start formulas. Most people think that you have to start a formula with the equal sign = but you can use the plus sign + to start formula. When you press Enter Excel will add the = at the front.

I use the large plus key on the numeric keypad on the right of the keyboard. This is a lot easier and quicker to hit than the = key, which is tucked away in the middle of the keyboard.

If you are using a laptop, without a numeric keypad, then this shortcut doesn't apply.

Numeric Keypad

I also use the numeric keypad for entering most of my operators for my calculations e.g. * / - + these are easier to hit on the numeric keypad, as are the numbers because the keypad uses the calculator layout and not the phone layout. Also the asterix (star) and plus sign + don't require the Shift key.

Using the Arrow Keys

Depending on the formula you are creating, it can be quicker to use the arrow keys to select cells than using the mouse or typing. We will see examples of this later in the session.

AutoComplete

Since Excel 2007 Excel has an AutoComplete feature that works with functions. As you type the function name, Excel will list the functions that start with the letters you have entered. You can select the selected function by pressing the tab key or by double-clicking with the mouse. You can use the arrow keys to select a function. You press the Tab key when the function you require is highlighted.

The advantage with using the AutoComplete feature is that it automatically inserts the opening bracket for the formula. This saves you having to use the Shift key and the bracket key.

In my training I have found that most people tend to ignore the AutoComplete, which is unfortunate because it saves you time and effort.

Last Bracket

Whilst the AutoComplete can enter the opening bracket for your functions, you can also avoid entering the last bracket for your function. Excel will automatically add the last bracket when you press Enter if you have included all of the arguments required by the function.


If you have made an error in the formula; or have missed out an argument in a function or an earlier bracket, Excel will typically display a dialog that shows you the formula that it is going to enter. You can accept or reject this dialog. If you reject it you can edit the formula to correct it.

AutoSum

The AutoSum icon is probably one of the most used icons on the ribbon. Many people have not noticed that it has a drop down.

This allows you to quickly select other common functions apart from the SUM function. See image on right.

The keyboard shortcut for the SUM function is **Alt + =**



F4 Key

When creating formulas it is common to insert \$ signs into the references to fix part of the reference so that it does not change as the formula is copied across and down.

The F4 function key is the easiest way to enter the \$s. Note the F4 function key works slightly differently between creating a formula and editing a formula. I will explain the differences during the session.

The \$ sign fixes whatever it is in front of.

The sequence of the F4 key presses is

A1 ==> \$A\$1 ==> A\$1 ==> \$A1 ==> A1

This sequence keeps repeating.

PRACTICE:

1. In the Quick Tips sheet create a SUM function in cell B6 using the AutoSum. Use Undo Ctrl + z and then use the Alt + = to create it. See **Warning** about AutoSum on the next page.
2. In cell C2 we will create a percentage calculation. Use the arrow keys to create the following formula. You can change the format to percentage.

=+B2/B6

3. This formula can't be copied down as the reference to the total in B6 will change. We need to amend B6 so it won't change as we copy it down.
4. Select cell C2 and press the F2 function key. This allows us to edit the formula. You can click on either side or in the middle of the B6 reference and press the F4 function. The formula below can be copied down.

=+B2/\$B\$6

5. When doing percentages it a good idea to make sure the sum adds up to 100%. Use another SUM function in cell C6 to calculate the total. Notice that the format for the cell automatically reverts to percentage.
6. Use cell B7 to calculate the average of the above tonnes. You can use the AutoSum icon or type it.
7. Use column E to calculate the value of the tonnes using columns B and D.
8. Use column F to calculate the GST on each row. Use a fixed reference for the GST cell G1.
9. In cell G11 create a formula than can be copied down and across to complete the table. You will need to use mixed cell references to complete this.

The final formula for cell G11 is

=+\$F11*G\$10

AUTOSUM WARNING: when you use AutoSum Excel guesses the range. Always check the range is correct. In most cases Excel will guess correctly, but blank cells will affect the range selected. You may need to manually adjust the range.

Copy Down

Mouse Shortcut

Copying down is a common practice for formulas. There is a mouse shortcut that can speed up this process. The more recent versions of Excel have a reasonably intelligent shortcut that reduces the risk of overwriting formulas.

Keyboard Shortcut

The keyboard shortcut to copy down is Ctrl + d. The formula you want to copy down should be at the top of the range and press Ctrl + d to populate all of the other cells with the copied cell. This works for entries as well as formulas.

Copy Across

The keyboard to copy across to the right is Ctrl + r.

Try the above shortcuts on the previous example in cell G11 of the Quick Tips sheet.

F2 Edit

The F2 function key edits the formula. This can also be used to remove a frustration when working with formula dialogs. When you try to use the arrow keys to move around the formula you actually select cells on the current sheet. This is frustrating. Pressing F2 removes that frustration and allows you to move around the formula using the arrow keys.

Calculation Sequence

When performing calculations in Excel you need to use the standard mathematical calculation sequence.

What is the answer to this equation?

$$5 + 10 * 2 = \underline{\hspace{2cm}}$$

The two typical responses are 30 and 25 – which is right?

In the Calculation Sequence sheet, use cell D1 to find out.

The abbreviated calculation sequence is BIMDAS

Brackets ()

Indices ^ (above numeral 6 on the main keyboard)

Multiplication * use numeric keypad

Division / use numeric keypad

Addition + use numeric keypad

Subtraction - use numeric keypad

Use cell D2 to enter this formula

$$=(A2+B2)*C2$$

Notice the different result.

Using Brackets

Brackets can be used to override the calculation sequence and make sure your calculations are performed in the sequence that you require.

Brackets Colour

Excel uses colour coding to match brackets. **Note:** when using a function with nested functions, the first bracket is always black which means the last bracket will also be black – if it isn't you may have missed a bracket.

Formula Colours

When you create and edit a formula you may have noticed that Excel applies colours to the sheet to display the references that are being used within the formula.

This is handy to make sure that the cells and ranges you are using are correct. What most people are unaware of is that you can use the colours to amend the formulas.

This becomes very useful when you are copying formulas around with fixed references because it becomes very easy to amend the formula with the mouse.

When you edit a formula the colours are visible on the sheet. When you point to the border of the colour you will see it become bold. You can point to the border with the mouse and then click, hold and drag the mouse to move or adjust the selected range.

If you point to the corner of the border you will see there is a double headed arrow which you can click hold and drag to amend the size of the range selected.

PRACTICE:

1. In the Colours sheet enter following formula in cell D3

`=+C3*D1`

2. Copy it down the column. Click one of the formulas press F2 and see the colours on the screen.
3. Copy cell D3 to H5. Press F2 and note the colours again. Point to the colour around cell D1 and watch it go bold. Click hold and drag the coloured cell to cell H3. You can now copy cell H5 down.
4. Repeat the process for the QLD report.
5. In the Colours_2 sheet enter the following formula in cell F2

`=SUMIF(A2:A17,E2,C2:C17)`

This conditionally adds up column C based on the entries in column A.

6. This can be copied down the column – try one of the shortcuts from earlier in the session.
7. Copy cell F2 to cell F9. Press the F2 function key and note the colours. These are wrong for our report but we can drag the column A range to column B to correct the formula. The final formula is

`=SUMIF(B2:B17,E9,C2:C17)`

Large Range Tips

Selecting large ranges in Excel can be time-consuming. There are a couple of techniques you can use which makes it easier to select larger ranges.

Whole Columns

If your range could expand then it can be worth referring to the whole column. This eliminates the need to expand the range as data is added to the range.

Using whole columns can affect calculation speed so use it sparingly.

Selecting Large Ranges

If you are selecting a range then you can use a keyboard shortcut to select a large range fairly quickly. This assumes that there are no blank cells within your range.

Simply select the first cell in the range and then hold the Ctrl and Shift keys down and press the down arrow. This will select the range from the cell you have selected to the very last cell in that column. If there are blanks in the range this technique will not be as effective.

Typing Shortcuts

If you need to select a large range which will handle extra data being added to the sheet rather than selecting the whole column you just may need to select a range that is large but at the moment doesn't have any data in it.

There is a quick and easy technique that allows you to enter a large range very quickly.

PRACTICE:

1. In the Large sheet in cell G2, enter the start of the formula as below

=SUMIF(

2. Use left arrow key to select cell C2. Then hold both the Ctrl and Shift keys down and press the down arrow.
3. Press F4, this fixes the whole range reference. Type a comma then use the left arrow key to select cell F2 and type another comma.
4. Use the left arrow key to select cell D2 and use the Ctrl + Shift + down arrow again and press F4 and then press Enter. Formula finished. It should look like the one below.

=SUMIF(\$C\$2:\$C\$547,F2,\$D\$2:\$D\$547)

5. If you need to select a large range that doesn't have any entries at the moment i.e. you want to select a range that, over time, will be populated with data. There is a relatively easy and


quick way to do this.

6. Delete the formula in G2 and start from scratch again with =SUMIF(
7. This time use the mouse or keyboard to select the range C2:C10 and press F4 to fix the range.
8. Then the trick is to type three zeroes. Which will create a fixed reference to \$C\$2:\$C\$10000.
9. Press the comma and use the same techniques to complete the formula below.

=SUMIF(\$A\$2:\$A\$10000,F2,\$D\$2:\$D\$10000)

Format As Table

One way to handle expanding ranges is to use the Format As Table option. There are many reasons to use Format As Table and I have included it in a couple of my other free webinars, so I won't go into all the benefits here. Suffice to say after applying the Format as Table option to a table you can use auto-expanding table names in your formulas.



PRACTICE:

1. In the Format As Table sheet click in the large table on the left side of the sheet. Press Ctrl + t and press Enter. This creates a Table with the default colour. The Format as Table icon (Home ribbon) provides the ability to select a colour.
2. In cell G2 enter the start of the formula as below

=SUMIF(

3. Use left arrow key to select cell C2. Note what the formula now has

=SUMIF(Table1[@State]

Press Ctrl + Shift + down arrow and the formula will be the same except the @ will not be there. Scroll back to the top of the sheet.

4. Type the comma and click cell F2 and type the comma and click cell D2, again use Ctrl + Shift + down arrow, note the formula. Press Enter.

```
=SUMIF(Table1[State],F2,Table1[Amount])
```


5. You can copy this formula down the column.
6. The @ symbol means refer to the field on the same row as the row you are on. In the above case we needed to refer to the column, not the cell in the column.

The correct term for this type of reference is a Structured Table reference. I call them table names. They are very similar to range names. See warning below about copying structured references.

Table1 is the generic name given to the table.

You can change it in the Design tab that appears when you select a cell in the table. See image on right.

Excel 2007 uses the # symbol instead of the @ symbol to identify field/column names.



This formula will cope with data added to the bottom of table.

TABLE WARNING: When you use the Fill Handle to copy formulas containing Structured References across left or right you will get unusual results. The references act as if they are relative references and change accordingly. Try copying G2 to H2 using the Fill Handle. Now try it using Copy and Paste – note the differences.

Include Inserted Rows

A common problem you can have when creating formulas, especially SUM functions, is that additional rows are inserted and these rows are not picked up by the SUM function. This is less of an issue in the more recent versions of Excel because Excel will look for inserted rows and they are, in many cases, automatically added to the SUM range.

One technique is to include a blank row at the bottom of the table and do the sum below the blank row – see image on right.

The formula in cell B7 is =SUM(B2:B6)

If a row or rows are inserted above row 6 they will automatically be added to the SUM range.

Try inserting rows above row 6 in the Inserted sheet and enter values in column B to see if they are included in the SUM.

	A	B
1	State	Amount
2	WA	7654
3	NSW	8765
4	VIC	4567
5	QLD	2345
6		
7	Total	23331

Try inserting rows above row 7 and see the result of entering numbers to column B.

In the cases where you want make sure Excel is always picking up the rows that have been inserted above the SUM function you can use the OFFSET function to make sure that the range is always finishing in the row above the SUM function.

Use Undo to remove the inserted rows from above so that the report look like the image above.

PRACTICE:

1. In the Inserted sheet enter the following formula in cell C7

=SUM(C2:OFFSET(C7,-1,0))

The OFFSET function provides a flexible reference to the cell above the current cell.

The -1 in the OFFSET function means to go up one row from C7 to supply the ending cell for the range. The 0 means not to change the column.

Inserting any rows between 2 and 7 will be automatically included in the SUM range.

NOTE: This won't handle rows inserted above row 2.

Other Sheet Ranges Tip

If you need to refer to ranges in other sheets then let me share with you a technique that I use all the time.

Let's take the SUMIFS function as an example because typically it refers to multiple ranges in another sheet.

Rather than selecting a range and then returning to the sheet with the formula what you can do is select all of the ranges whilst you are in the source sheet. Rather than using the actual references you will use in the final formula you can just insert place holders that you can replace when you return to the final sheet.

Let's have a look at the two techniques to create the formulas. First I'll show you the normal technique that people use which involves selecting the reference sheet multiple times.

Then I'll show you my shortcut technique and you can decide which one is easier to use. I will refer to whole columns in both examples to speed up the process a little bit. It takes time master, but it does speed up formula entry.

PRACTICE:

1. In the Report sheet in cell B3 following these steps

Enter =SUMIFS(

Click the Data sheet and select column D and press F4 type a comma

Select column C and press F4 then type a comma

Click the Report sheet tab and click cell A3 and press the F4 key three times to fix the column reference

Click back in the Data sheet and select column A and press F4 type a comma

Then type ">="& and then click the Report sheet and click cell B1 and press the F4 key twice to fix the row reference, type a comma

Click back in the Data sheet and select column A and press F4 type a comma

Then type "<="& and then click the Report sheet and click cell B2 and press the F4 key twice to fix the row reference press Enter.

The final formula (can be copied down and across) is shown below

=SUMIFS(Data!\$D:\$D,Data!\$C:\$C,Report!A3,Data!\$A:\$A,">="&
Report!B\$1,Data!\$A:\$A,"<="&Report!B\$2)

Note that Excel has included the sheet name Report! in each of the cells on the Report sheet. The sheet names are redundant are not required and can be removed. The technique below does not have that issue.

2. Compare the above technique with this one

Enter =SUMIFS(

Click the Data sheet and select column D and press F4 type a comma

Select column C and press F4 to fix it

Type a comma then type a 1 then another comma

Click column A and press F4 and type a comma and type a 2 and another comma

Click column A and press F4 and type a comma and type a 3 and press Enter.

The formula is not complete, but all the references to the other sheet are captured.

=SUMIFS(Data!\$D:\$D,Data!\$C:\$C,1,Data!\$A:\$A,2,Data!\$A:\$A,3)

Now edit the formula and replace the 1, 2 and 3 as shown below in the final formula.

=SUMIFS(Data!\$D:\$D,Data!\$C:\$C,\$A3,Data!\$A:\$A,">="&B\$1,
Data!\$A:\$A,"<="&B\$2)

Once you get the hang of this technique – it does take practice and knowledge of the structure of the data, it becomes quite quick to create formulas referring to other sheets.

You can copy B3 to B7 to create a check total as it uses the *.

Conclusion

I have shared lots of tips, tricks and techniques to help you create formulas and functions quickly and easily. Many of these require practice to master, so identify those that are most worthwhile and start implementing them in your Excel work. As you improve and they become second nature, add other techniques.

If you don't use it you lose it, so practice and incorporate these lessons into your daily Excel practice.